NCSX

Design Basis Analysis

Deformations in the NCSX TF Coil from all Field Sources and a Simplistic Linear Model

NCSX-CALC-131-004-00

03 January 2005

Prepared by:

____

L. Myatt, Engineering Analyst

I have reviewed this calculation and, to my professional satisfaction, it is properly performed and correct. I concur with analysis methodology and inputs and with the reasonableness of the results and their interpretation.

Reviewed by:

Art Brooks

Digitally signed by Art Brooks
DN: cn=Art Brooks, o=PPPL, ou=NCSX, email=abrooks@pppl.gov, c=US
Date: 2008.10.01 07:52:52 -04'00'

____

A. Brooks, Engineering Analyst

1.0 Executive Summary

This memo describes an ANSYS[1] electromagnetic-structural analysis of the NCSX coil systems. The 3D model is fully developed in an earlier project memo[2], and so the text is intentionally brief.

The objective of the analysis is to determine the TF coil displacements from some of the worst-case EM loading, such as the TF at 0.5 T, 1.7 T Ohmic and 2.0 T High-β. Displacements are presented in plot form for visualization purposes, but also as text files for ease of use by others (but not attached to this memo).

It is worth noting that this latest version of the memo includes results for the 3x4 TF coil WP with a 3/8" thick ground wrap; updated smeared orthotropic material properties are used[3]. In addition, the global model includes the reference 4000 lb/bracket radial preload which affects deflections (albeit symmetrically).

Results indicate that the 0.5 T operating case continues to be the design-basis loading for the stress in the TF coil system. Field error analysis may show that other loadings with more out-of-plane deformations provide a more limiting condition for field criteria.

Results from the 2x6 conductor array design are maintained for a historical reference even though the design described above has been officially adopted by the project. So, consider paging directly to section 3.3 to avoid reviewing out-of-date analyses.

---

[1] ANSYS Release 8.1, UP20040329, INTEL NT, ANSYS, Inc., Canonsburg, PA.
[2] Leonard Myatt, "Electromagnetic Stress Analysis of the NCSX TF and PF Coil Systems," 09/24/03.
[3] Leonard Myatt, "Calculating Smeared Properties of the TF Winding Pack for Use in Global Models," January 3, 2005.

2.0 Analysis

The stress analysis of the NCSX TF & PF coil systems is based on the finite element ANSYS model shown in Fig. 2.0-1. The coupled electromagnetic-structural model is an accumulation of a number of references:

- The coil support structure (shown in red) is a defeatured version of the PPPL solid model[4].
- Coil geometries (TF, PF, Modular and Plasma) are from a PPPL ANSYS database[5].
- Coil currents for various operating scenarios are from the NCSX Technical Data site[6].
- Smear material properties for PF[7] & TF[8] winding packs.



Fig. 2.0-1: NCSX Magnet System 120˚ Sector

---

[4] T. Brown, sjdr130-003-c4-fea_asm.igs (Structural members, 09/02/03), se151-019.igs (PF6 bracket, 09/10/03).
[5] H.M. Fan, ANSYS database from ftp://ftp.pppl.gov/pub/hmfan/ncsx-pftf/file-1.7T.db
[6]http://ncsx.pppl.gov/NCSX_Engineering/Requirements/Specs/GRD/Rev1/TDS_XL_C08R00_c3.pdf
[7] Chang Jun, "ANSYS Modeling to obtain Equivalent Moduli of Elasticity of PF & TF Coils of NCSX," 06/27/03
[8] Chang Jun, "ANSYS Modeling to obtain Equivalent Moduli of Elasticity of TF Coils of NCSX," 07/15/03

These inputs are manipulated into a text-based ANSYS batch file (included here as Attachment 4.1) which facilitates changes such as material properties, boundary conditions, coil currents and postprocessing with minimal user-input. The model is well developed in [2] and the interested reader is referred there for details relating to boundary conditions, load application and material properties. All of these modeling details are available by way of the attached text-based ANSYS input file. However, some knowledge of the ANSYS language might be required to decipher the coding.

A series of figures are included here to illustrate the TF coil support. Fig. 2.0-2 shows how shims are placed between the coil pack and the support structure. Fig. 2.0-3 shows how the displacement continuity between the misaligned WPs and shims are handled with constraint equations. Fig. 2.0-4 shows how the TF coils are supported from below. Fig. 2.0-5 shows how the TF support structure is fixed in space as if attached to an infinitely rigid Modular Coil shell.

Fig. 2.0-2 TF Winding-Pack and Support Structure Lateral Continuity



Fig. 2.0-3 TF Coil-Shim Constraint Equation Symbols



Fig. 2.0-4 TF Coil Vertical Displacement Constraints



Fig. 2.0-5 Coil Support Structure Anchored to Modular Coil Shell at Bolted Flanges





Inboard Structure and Bolt Location



Outboard Structure and Bolt Location

Fig. 2.0-6 represents the latest design modification to the TF coil system. Here, the toroidal dimension of the TF WP beyond the wedged region is 3.468”. However, in the wedged region, the WP is expanded to produce a toroidally continuous (wedged) structure through a 50˚ arc beyond the straight inboard leg. This simplistic representation of the TF coil and structure does not specifically include the SS wedges as in the more precise model[9] shown in Fig. 2.0-7. However, it is a realistic representation of the structure’s stiffness, and a fair model for determining deflections for field error studies.

[9] Leonard Myatt, “Stress Analysis of the Narrow (3.47”) NCSX TF Coils with Cast SS Wedges,” 10/19/2004.

Fig. 2.0-6 Reduced-Width (3.47" Wide) TF Coil and Wedged Region Extended by ~50˚



Fig. 2.0-7 Precise Model of 3.47" Wide TF Coil and Wedged Region (taken from [9])



3.0 Analysis

The finite element model described above is used to simulate what is believed to be some of the most limiting Load Cases (LC) for deflection-induced field errors. The list includes the TF coil at 0.5 T, 1.7 T Ohmic, 2.0 T High-β and Equilibrium ID 7 (the latter has the TF at full current plus Modular Coils at close to their full current [6]). Although it is not part of the reference scenarios [6], this equilibrium operating condition probably produces the largest out of plane loads and displacements for the TF coils.

The coil currents associated with these Load Cases and number of turns per coil, N, are listed in Table 3.0-1. The steady-state 80K, zero-current, operating condition is referred to as Load Case 0 in the table. Tables 3.0-2 to 3.0-4 are cut from the Project's Web site on Technical Data [6] and form the basis of the analyzed operating conditions.

The focus of this analysis is TF coil displacements which serve as input to field error calculations. However, some stress contour plots are also presented for casual review only. With an element density of 2x2, the WP cross-section is far too coarse to believe that the model is capable of producing accurate bending stresses.

Table 3.0-1 Summary of Currents [A] for Limiting Load Case

Load Case	0	1	2	3	6
Ref. Run #	All	25	26	27	28
Description	80K Only	TF @ 0.5 T	1.7 T Ohmic	2.0 T High-β	EQ ID 7
Time, s	0	N/A	0.0	0.0	N/A
M1 (N=20)	0	0	38141	40908	34750
M2 (N=20)	0	0	35504	41561	35300
M3 (N=18)	0	0	35453	40598	34500
PF1 (N=72)	0	0	-25123	-15274	0
PF2 (N=72)	0	0	-25123	-15274	0
PF3 (N=72)	0	0	-9698	-5857	2222
PF4 (N=80)	0	0	-7752	-9362	-2400
PF5 (N=24)	0	0	8284	1080	1008
PF6 (N=14)	0	0	-8997	-24	764
TF (N=12)	0	16200	-3548	-1301	16200
Plasma (N=1)	0	0	0	0	0
Displacement File tflocdisp*.lis	25LC0LS1	25LC1LS2	26LC2LS2	27LC3LS2	28LC6LS2

Note: In ANSYS vernacular, Load Step 1 (LS1) of each analysis (LC 1, 2, 3 and 6) corresponds to the so-called 0 load case. Load Step 2 of these analyses includes both thermal and EM loads. The Load Step number is shown in the ANSYS plot legend.

Table 3.0-2 Current Wave Form, 1.7 T Ohmic Scenario

1.7T Ohmic Scenario	t(s)	M1	M2	M3	PF1	PF2	PF3	PF4	PF5	PF6	TF	Plasma
	-0.700	0	0	0	0	0	0	0	0	0	0	0
	0.000	38141	35504	35453	-25123	-25123	-9698	-7752	8284	-8997	-3548	0
	0.100	38141	35504	35453	-25123	-25123	-9698	-7752	8284	-8997	-3548	0
	0.140	34772	35327	34508	-16185	-16185	-6754	-9327	743	4396	-1106	-120052
	0.240	34772	35327	34508	-12683	-12683	-4812	-7829	766	4609	-1106	-120052
	0.440	34772	35327	34508	-5681	-5681	-928	-4832	814	5033	-1106	-120052
Maximum		38141	35504	35453	0	0	0	0	8284	5033	0	0
Minimum		0	0	0	-25123	-25123	-9698	-9327	0	-8997	-3548	-120052
I2t (A2-s)		1.39E+09	1.29E+09	1.26E+09	4.24E+08	4.30E+08	6.27E+07	4.95E+07	4.61E+07	6.36E+07	1.05E+08	
tESW (s)		0.96	1.02	1.00	0.67	0.68	0.67	0.57	0.67	0.79	8.31	

Table 3.0-3 Current Wave Form, 2.0 T High-β Scenario

2T High Beta Scenario	t(s)	M1	M2	M3	PF1	PF2	PF3	PF4	PF5	PF6	TF	Plasma
	-0.850	0	0	0	0	0	0	0	0	0	0	0
	0.000	40908	41561	40598	-15274	-15274	-5857	-9362	1080	-24	-1301	0
	0.050	40908	41561	40598	-15274	-15274	-5857	-9362	1080	-24	-1301	0
	0.097	40908	41561	40598	-7612	-7612	-1607	-6082	952	5865	-1301	-141238
	0.192	38583	38338	35315	-8219	-8219	-2857	-8701	2666	7033	4424	-209732
	0.197	38583	38338	35315	-8201	-8201	-2848	-8693	2666	7034	4424	-209732
Maximum		40908	41561	40598	0	0	0	0	2666	7034	4424	0
Minimum		0	0	0	-15274	-15274	-5857	-9362	0	-24	-1301	-209732
I2t (A2-s)		1.51E+09	1.54E+09	1.43E+09	1.68E+08	1.68E+08	2.41E+07	7.08E+07	2.82E+06	7.40E+06	2.25E+08	
tESW (s)		0.90	0.89	0.87	0.72	0.72	0.70	0.81	0.40	0.15	11.50	

Table 3.0-4 Current Wave Form, Equilibrium ID 7

A.3.1 Reference Equilibria

Equlibrium ID	M1	M2	M3	PF1	PF2	PF3	PF4	PF5	PF6	TF	Plasma		Comment
1	7.63E+05	7.10E+05	6.38E+05	0.00E+00	0.00E+00	3.05E+05	2.40E+05	2.03E+05	-1.05E+05	-4.26E+04	0		iota>0.5
2	6.95E+05	7.06E+05	6.21E+05	0.00E+00	0.00E+00	1.60E+05	-1.92E+05	2.42E+04	1.07E+04	-1.33E+04	0		iota<0.5
3	6.95E+05	7.06E+05	6.21E+05	0.00E+00	0.00E+00	1.60E+05	-1.92E+05	2.05E+04	7.53E+04	-1.33E+04	-120000		120kA, zero beta
4	6.59E+05	6.54E+05	5.43E+05	0.00E+00	0.00E+00	1.05E+05	-3.54E+05	5.58E+04	9.00E+04	4.53E+04	-179000		179kA, full beta
5	6.82E+05	6.40E+05	5.78E+05	0.00E+00	0.00E+00	-1.30E+06	-1.50E+06	1.07E+05	6.12E+04	2.62E+04	-320000		320kA, zero beta
6	6.69E+05	6.44E+05	5.57E+05	0.00E+00	0.00E+00	-1.14E+05	-2.09E+05	-3.27E+05	2.60E+05	3.77E+04	-160000		160kA, zero beta
7	6.95E+05	7.06E+05	6.21E+05	0.00E+00	0.00E+00	1.60E+05	-1.92E+05	2.42E+04	1.07E+04	1.94E+05	0		0.5T TF +S1b
8	7.63E+05	7.10E+05	6.39E+05	0.00E+00	0.00E+00	0.00E+00	9.64E+05	0.00E+00	-9.72E+03	-4.26E+04	0		First Plasma
9	6.95E+05	6.98E+05	6.29E+05	0.00E+00	0.00E+00	0.00E+00	-8.17E+04	0.00E+00	8.42E+04	-1.33E+04	-120000		First Plasma

3.1 Deformations in the Original (2x12) TF WP, Equilibrium ID 7

Contours of the displacement modulus are shown in Figs. 3.1-1 through 3.1-3 for the original 2x6 turn WP. Fig. 3.101 is a plot of the deformations associated with cooldown to 80K. Notice that the smallest displacements are on the bottom. Recall Fig. 2.0-4 which shows that these coils are supported from the bottom. The plot shows that the upper-outer part of the coil moves down and in about 8 mm from thermal contraction effects.

Fig. 3.1-2 is a plot of the displacements in the original 2x12 WP when the coil system is at 80K and currents are at EQ ID7 levels. In this case the maximum motion is the outboard leg moving inward by ~19 mm (or ¾"). When the deflections from thermal effects are subtracted out of the thermal plus EM load case, the results are as plotted in Fig. 3.1-3. Both cooldown and cooldown + EM displacement data are made available in ASCII text form at TF coil element centers for ease of input to other field error programs.

Fig. 3.1-4 is a plot of the stresses associated with this EQ ID7 loading. Again, the FE mesh is very coarse, and is not likely to be very accurate except serendipitously. It is presented here to be used on a comparative basis.

Fig. 3.1-1 TF Coil Displacements from Cool Down Only



Fig. 3.1-2 TF Coil Displacements from Cool Down + Equilibrium ID #7 EM Loads



Fig. 3.1-3 TF Coil Displacement Differential (Cool Down + Eq ID #7) – (Cool Down)



Fig. 3.1-4 TF Coil Stress Intensity (Cool Down + Eq ID #7)



3.2 Deformations and Stresses in the New (3x4) TF WP, 0.12" GW

Contours of the displacement modulus (with deformations exaggerated 100x for visual impact) and stress are shown in Figs. 3.2-1 through 3.2-8 for the new 3x4 turn WP. As before, displacement data at TF coil element centers is made available in plain text formatted files for ease of input to other field error programs, and the stress contour plot should be used for comparative purposes only.

Figs. 3.2-1 and 3.2-2 are plots of the displacement modulus and stress intensity from the LC 1 (TF at 0.5 T) operating currents. The maximum deflection is 7.8 mm and the maximum stress is 33 MPa (5 ksi). Both plots show perfect coil to coil symmetry, as expected with this "TF Only" loading. It is no great surprise that the stresses predicted by this coarse model are not very close to the 93 MPa results reported by a much more detailed model [9] for this particular loading.

Figs. 3.2-3 and 3.2-4 are plots of the displacement modulus and stress intensity from the LC 2 (t=0s of the 1.7 T Ohmic scenario). The maximum deflection is 8.7 mm and the maximum stress is 17 MPa (2.5 ksi).

Figs. 3.2-5 and 3.2-6 are plots of the displacement modulus and stress intensity from the LC 3 (t=0s of the 2.0 T High-β scenario). The maximum deflection is 8.3 mm and the maximum stress is 7 MPa (1 ksi).

Figs. 3.2-7 and 3.2-8 are plots of the displacement modulus and stress intensity from the LC 6 (Equilibrium ID7). The maximum deflection is 10.9 mm and the maximum stress is 93 MPa (13 ksi).

Fig. 3.2-1 Displacement Modulus [m] in 3x4 TF WP from LC1 (TF only at 0.5 T)



Fig. 3.2-2 Stress Intensity [MPa] in 3x4 TF WP from LC1 (TF only at 0.5 T)



Fig. 3.2-3 Displacement Modulus [m] in 3x4 TF WP from LC2 (1.7 T Ohmic)



Fig. 3.2-4 Stress Intensity [MPa] in 3x4 TF WP from LC2 (1.7 T Ohmic)



Fig. 3.2-5 Displacement Modulus [m] in 3x4 TF WP from LC3 (2.0 T High-β)



Fig. 3.2-6 Stress Intensity [MPa] in 3x4 TF WP from LC3 (2.0 T High-β)



Fig. 3.2-7 Displacement Modulus [m] in 3x4 TF WP from LC6 (EQ ID7)



Fig. 3.2-8 Stress Intensity [MPa] in 3x4 TF WP from LC6 (EQ ID7)



Notable Concern:

The analysis is designed to produce TF Coil deformations as a result of the most significant EM loads. However, it uses a very simplistic, linear model (Fig. 2.0-6) to represent a more complex structure (Fig. 2.0-7). Fig. 3.2-9 is a plot of the deformations resulting from the TF coils at 0.5 T. It should be noted that run #25 is composed of two Load Steps; (1) cooldown to 80 K, (2) TF coils carrying 16200 A x 12 turns (at 80K). In addition, the coils are supported from the bottom.

In order to compare the deflections produced by this approximate model with those of the more detailed model, Load Step 1 results are subtracted from Load Step 2 results. This leaves just EM effects in the "data base" for postprocessing. Next, the effects of different vertical support locations are eliminated. The vertical displacement (UZ) at the midplane (z=0) is subtracted from the database results. Finally, a displacement modulus (U) is produced by performing a SRSS on the three displacement components.

The plot shown below indicates that the coil moves vertically about 1 mm at the top and bottom, and radially about 0.8 mm at the outboard leg. Compare this with the TF Only results from [9] shown in Fig. 3.2-10.



Fig. 3.2-9 Displacements of a TF Coil with Coarse-Mesh Linear Model (0.5 T EM, No Thermal)

Although the more detailed model produces comparable results at the bottom (1.0 mm vs. 1.2 mm) where the wedges are bonded to the WP (form modeling efficiency), the disparity is substantial at the top (1.0 mm vs. 2.9 mm) because of the intricacies of the contact surfaces and local deformations of the wedge. This effect also appears in outboard leg deformations which are -0.8 mm in the linear model and -1.8 mm in the more detailed nonlinear model.

One approach at reconciling this modeling shortcoming would be to use the more detailed model TF Coil model in the frame work of a global simulation. (So far, the detailed TF model has only been used to simulate TF fields and forces.) This would probably be a computational nightmare, but could be attempted. Another option would be to simply scale the nonconservative deformations of the simplistic model by say 2.9/1.0 (2.9) or 1.8/0.8 (2.3). Obviously, this would be a very simple approach, albeit marginally defendable.



Fig. 3.2-9 Displacements of a TF Coil with Nonlinear Hybrid Model [9] (0.5 T EM, No Thermal)

3.3 Deformations and Stresses in the New (3x4) TF WP, 0.375" GW

The use of this simplified, linear global model to capture the deflections in the structurally complex TF coil system requires some approximations. Here, each coil and associated wedges are combined into one monolithic smear which is "bonded" to its neighbor. This approach generally leads to a structure which is stiffer than reality; general contact is never as stiff as a bond.

The reference design shows the cast wedges extending ~50˚ beyond the inboard straight-leg (see Fig. 2.0-7). This design detail is applied to the global model as shown in Fig. 2.0-6. However, the multi-contact surface TF coil and wedging structure is actually *softer* than the linear global model would suggest. The strongest parameter in determining the stiffness of the structure is the wedge-dewedge transition angle. Large transition angles (like the reference 50˚ value) produce a stiff structure while small transition angles can be used to reduce the radial stiffness of the coil structure.

A short study is used to determine the extent of the transition region which best simulates the stiffness of the actual coil system. Thermal contraction effects to 80 K are ignored. Fig. 3.3-1 shows a plot of the radial deformations of the TF coil hybrid model. The wedges and insulation are unselected leaving only the smeared WP and detailed conductors. The blue contours indicate that the maximum radial deformation from this 0.5 T load case is -2.8 mm and occurs in the outboard leg, slightly off of the equatorial plane.



Fig. 3.3-1 Radial Deformations in the Hybrid Model TF WP and conductor from 0.5 T Loads

Figs. 3.3-2 and 3.3-3 are also radial displacement plots from the 0.5 T load case, but both come from linear global models with dewedge regions at 45˚ and 40˚ beyond the inboard leg. When the de-wedging occurs 45˚ from the straight leg, the maximum radial deformation is -2.5 mm (Fig. 3.3-2). When the de-wedging occurs 40˚ from the straight leg, the maximum radial deformation is -3.1 mm (Fig. 3.3-3). Based on these results, the global model behaves most like the hybrid model when the effective transition region is right between 40˚ and 45˚. Unfortunately, the poloidal dimension of the TF elements does not allow this configuration. So, we choose the more conservative approximation, 40˚, for the remaining studies.



Fig. 3.3-2 Radial Deformations in the Global Model TF WP from 0.5 T Loads (45˚ Dewedge)



Fig. 3.3-3 Radial Deformations in the Global Model TF WP from 0.5 T Loads (40˚ Dewedge)

With the linear global model stiffness “tuned” to more closely resemble that of the hybrid model, we are now prepared to add the 4000 lb radial preload and determine TF coil deformations for the three most important load conditions: 0.5 T TF Only, 1.7 T Ohmic Heating, and 2.0 T High-β. It is also worth noting that the TF coil smeared properties are consistent with this 3x4 conductor array and 3/8” ground wrap (from [3]).

Contours of the displacement modulus (with deformations exaggerated 100x for visual impact) and stress are shown in Figs. 3.3-4 through 3.2-9. As before, displacement data at TF coil element centers is made available in plain text formatted files for ease of input to other field error programs. A significant portion of the deflections are due to thermal contraction, which tends to mask the EM load component. We really rely on field error calculations to evaluate these deformations. Stress contour plot should be used for comparative purposes only since *unsmearing* stresses is a questionable methodology.

Figs. 3.3-4 and 3.3-5 are plots of the displacement modulus and stress intensity from the LC 1 (TF at 0.5 T) operating currents. The maximum deflection is 11.6 mm and the maximum stress is 58 MPa (8.5 ksi). Both plots show perfect coil to coil symmetry, as expected with this “TF Only” loading.

Figs. 3.3-6 and 3.3-7 are plots of the displacement modulus and stress intensity from the LC 2 (t=0s of the 1.7 T Ohmic scenario). The maximum deflection is 11.5 mm and the maximum stress is 31 MPa (4.5 ksi).

Figs. 3.3-8 and 3.3-9 are plots of the displacement modulus and stress intensity from the LC 3 (t=0s of the 2.0 T High-β scenario). The maximum deflection is 10.8 mm and the maximum stress is 22.7 MPa (3.3 ksi).

It should be noted that the 0.5 T operating case continues to be the design-basis loading for the stress in the TF coil system. Field error analysis may show that other loadings with more out-of-plane deformations provide a more limiting condition for field criteria.

Fig. 3.3-4 Displacement Modulus [m] in 3x4 TF WP from LC1 (TF only at 0.5 T)



Fig. 3.3-5 Stress Intensity [MPa] in 3x4 TF WP from LC1 (TF only at 0.5 T)



Fig. 3.3-6 Displacement Modulus [m] in 3x4 TF WP from LC2 (1.7 T Ohmic)



Fig. 3.3-7 Stress Intensity [MPa] in 3x4 TF WP from LC2 (1.7 T Ohmic)



Fig. 3.3-8 Displacement Modulus [m] in 3x4 TF WP from LC3 (2T High-β)



Fig. 3.3-9 Stress Intensity [MPa] in 3x4 TF WP from LC3 (2T High-β)



4.0 Attachments

4.1 Batch Input File (circa 01/03/05)

```
/BATCH
rn=40
/filnam,pdrcoils1%rn%
/show,pdrcoils1%rn%,grp
!resume
!*if,1,eq,1,:1001
/PREP7

/com
/com EM-STRUC Analysis of the NCSX PDR TF/PF Coil System
/com
/com
/com  Run History
/com 10: TF Only, 16.2kA, 0.5T, Isotropic resistivity produces best results
/com 11: timepoint=1 (0.5 T TF Only), with CEs on TF sides
/com 12: timepoint=2 (1.7T Ohmic Scenario, t=0.0)
/com 13: timepoint=3 (2T High Beta Scenario, t=0.0)
/com 14: timepoint=4 (320kA Ohmic Scenario, t=0.206)
/com 15: TBD: timepoint=5 (320kA Ohmic Scenario, t=0.506)
/com Set all alpha values to 10e-6
/com 16: timepoint=1 (0.5 T TF Only, 85K), tunif=-215, tref=0, hold PF5/6 by CE
/com 17: timepoint=1 reft=300, bf=85, d,all,temp,85, hold PF5/6 w/CE (Structure stress not right)
/com 18: Just Cold, tref=0, bfunif=-215, d,all,temp,-215, hold PF5/6 w/CE, cnvtol,f,big
/com 19: timepoint=1 tref=0, bfunif=-215, d,all,temp,-215, hold PF4/5/6 w/CE, cnvtol,f,big

/com 11: Just Cold (0.5e-4 T TF), now with CEs on TF sides...looks OK

/com  Now with CEs on TF sides

/com 12: TP=1 (85K and 0.5 T TF)...looks OK
/com 13: TP=2
/com 14: TP=3
/com 15: TP=4
/com 16: TP=5

/com Incorporate the CS into the structural model

/com 111: TP=1
/com 112: TP=2
/com 113: TP=3
/com 114: TP=4
/com 115: TP=5

/com
/com June 2004, Analyze another load case to determine TF coil deformations for Art's field error
study
/com
/com 116: Equilibrium ID #7 has high TF and MC currents
/com 117: Thermal Only

/com
/com Sept 2004, Reduce width of TF WP to 3.47"
/com   Look for the logic that changes the WP width and the WP/Shim CEs
/com
/com 118: TF Only at 0.5T (tp=1)
/com 119: EQ ID #7 (tp=6)
/com 120: 2T high-Beta, t=0 (tp=3)
/com 121: 1.7T Ohmic, t=0 (tp=2)

/com Switch back to full-width WP for rerun of LC 2 & 3 (old analyses might have had too much NI in
MCs)
/com 122: 1.7T Ohmic, t=0 (tp=2)
/com 123: 2T high-Beta, t=0 (tp=3)
/com 124: 0.5T (tp=1)
```

```
/com
/com Oct 2004, Reduce width of TF WP to 3.47" ~50 deg around dewedging corner
/com   Look for the logic that changes the WP width and the WP/Shim CEs
/com
/com 125: TF Only at 0.5T (tp=1)
/com 126: 1.7T Ohmic, t=0 (tp=2)
/com 127: 2T high-Beta, t=0 (tp=3)
/com 128: EQ ID #7 (tp=6)

/com
/com Dec 2004, Revise TF WP smeared Properties and increase GW to 0.25"
/com
/com 129: TF Only at 0.5T (tp=1)
/com 130: 1.7T Ohmic, t=0 (tp=2)
/com 131: 2T high-Beta, t=0 (tp=3)

/com 132: 2T high-Beta, t=0 (tp=3), 0.75x softer Ey to simulate too big drwp to help fix
displacement underestimation
/com 133: TF Only at 0.5T   (tp=1), 0.75x softer Ey to simulate too big drwp to help fix
displacement underestimation
/com 134: TF Only at 0.5T   (tp=1), 0.75x softer Ey, 30 deg wedging
/com 135: TF Only at 0.5T   (tp=1), 0.75x softer Ey, 40 deg wedging

/com Late Dec 2004, Revise TF WP smeared Properties and increase GW to 3/8" GW
/com 136: TF Only at 0.5T   (tp=1), 45 deg wedging
/com 137: TF Only at 0.5T   (tp=1), 40 deg wedging
/com 138: TF Only at 0.5T   (tp=1), 40 deg wedging, add radial preload (4000 lb x 2 locations per TF
coil)
/com 139:   1.7T Ohmic, t=0 (tp=2), 40 deg wedging, add radial preload (4000 lb x 2 locations per TF
coil)
/com 140: 2T high-Beta, t=0 (tp=3), 40 deg wedging, add radial preload (4000 lb x 2 locations per TF
coil)

/com
/com Field Analysis based on coil definitions in HM Fan's file-1.7t.db (saved as pdrcoils1.db)
/com
!resume,pdrcoils1,db,,1      ! pull in PF/TF/MC/Plasma Current Sources and merge parameters
!resume,pdrcoils2,db,,1      ! pull in PF/TF/MC/Plasma Currents (Cheat: PF4 moved 1.12 cm to Struc
Surface)
resume,pdrcoils3,db,,1      ! pull in CS/PF/TF/MC/Plasma Currents (Add spacers between coils in CS)

/com
/com Misc Parameters
/com
time_pt=3                   ! Coil Currents for this time point (see array listed below)
f_preload=4000/0.2248       ! applied radial preload per TF Bracket
t=1e-3                      ! tiny
k=0.0254                    ! m/in
emsym,3                     ! number of circular symmetry sections about Global Z
r0=1.4                      ! Major Radius
tmp=-215                    ! Operating Temp
t_ref=0                     ! Reference Temperature
alpha_struc=10e-6           ! Thermal expansion coefficient of Structure (should be 13e-6 for SS at
85K)
tref,t_ref                 ! Or use MP command
!tunif,tmp                  ! Steve says do not use tunif

/pbc,mag
/pbc,temp
/pbc,volt
/pbc,ce

/com
/com Element Types
/com
et,1,5,0                    ! U,TEMP,VOLT,MAG (PF4/5/6 and TF)
et,2,36                     ! Plasma Current Source Primitive
et,3,5,1                    ! TEMP,VOLT,MAG (CS, M and R0 ring)

/com
/com Merge TF Coil nodes in the inboard leg
```

```
/com
esel,s,mat,,2
nsle
csys,1
nsel,r,loc,x,,r0
numm,node

/com
/com Material Properties
/com
allsel
mpdele,all,all
/com All PF Coils
!mp,reft,1,t_ref
mp, kxx,1,1
mp,murx,1,1
mp,rsvx,1,1
mp,alpx,1,10e-6    ! roughly (no reference or calc)
/com Chang Jun, "ANSYS Modeling to obtain Equivalent Moduli of Elasticity of
/com              PF & TF Coils of NCSX," 06/27/03
mp,  ex,1,62.27E9!,54.44E9
mp,  ey,1,93.10E9!,85.10E9
mp,  ez,1,64.03E9!,56.59E9
mp, Gxy,1,35.27E9!,32.01E9
mp, Gyz,1,35.27E9!,32.01E9
mp, Gxz,1,20.69E9!,18.31E9
mp,nuxy,1,  0.306!,0.310
mp,nuyz,1,  0.213!,0.209
mp,nuxz,1,  0.339!,0.340

/com TF Coils
!mp,reft,2,t_ref
mp, kxx,2,1
mp,murx,2,1
mp,rsvx,2,1
!mp,alpx,2,10e-6    ! roughly (no reference or calc)
/com Chang Jun, "ANSYS Modeling to obtain Equivalent Moduli of Elasticity of
/com              TF Coils of NCSX," 07/15/03
!mp,  ex,2,111.6E9!, 93.3E9
!mp,  ey,2,129.5E9!,119.0E9
!mp,  ez,2,125.1E9!,113.9E9
!mp, Gxy,2, 48.6E9!, 44.4E9
!mp, Gyz,2, 48.6E9!, 44.4E9
!mp, Gxz,2, 41.8E9!, 37.5E9
!mp,nuxy,2,  0.323!,  0.329
!mp,nuyz,2,  0.277!,  0.273
!mp,nuxz,2,  0.277!,  0.273
/com
/com From Myatt's Smear Properties Memo (3x4 TF with Kapton Slip-Plane and 3/8" GW)
/com
mp,alpx,2,9.5e-6
mp,alpy,2,13.0e-6
mp,alpz,2,9.6e-6
mp,  ex,2,42.3E9
mp,  ey,2,76.6E9
mp,  ez,2,41.2E9
mp, Gxy,2,2.1E9
mp, Gyz,2,2.1E9
mp, Gxz,2,2.1E9
mp,nuxy,2,0.284
mp,nuyz,2,0.284
mp,nuxz,2,0.319

/com M1 Coils
!mp,reft,3,t_ref
mp, kxx,3,1
mp,murx,3,1
mp,rsvx,3,1
/com M2 Coils
!mp,reft,4,t_ref
mp, kxx,4,1
```

```
mp,murx,4,1
mp,rsvx,4,1
/com M3 Coils
!mp,reft,5,t_ref
mp, kxx,5,1
mp,murx,5,1
mp,rsvx,5,1
/com R0 Field Elements
!mp,reft,7,t_ref
mp, kxx,7,1
mp,murx,7,1
mp,rsvx,7,1

/com CS Coil-Coil shims (G-10)
!mp,reft,9,t_ref
mp,murx,9,1
mp,  ex,9,25e9
mp,alpx,9,10e-6
mp,nuxy,9,.21
/com Coil Support Structure (Stainless Steel)
!mp,reft,10,t_ref
mp,murx,10,1
mp,  ex,10,205e9
mp,alpx,10,alpha_struc
mp,nuxy,10,0.265
mp, gxy,10,81e9
/com TF Coil-Case side shims (G-10)
!mp,reft,11,t_ref
mp,murx,11,1
mp,  ex,11,25e9
mp,alpx,11,10e-6
mp,nuxy,11,.21

/com
/com Current Nodes
/com
n_pfcoils=6   ! number of PF Coil Pairs
n_tfcoils=6   ! number of PF Coils
n_mcoils=3    ! number of Modular Coil Pairs
/com PF Coil Current Nodes
n_pf1u=1
n_pf1l=1488
n_pf2u=1110
n_pf2l=1593
n_pf3u=1299
n_pf3l=2346
n_pf4u=190
n_pf4l=1782
n_pf5u=450
n_pf5l=1922
n_pf6u=846
n_pf6l=2126
/com TF Coil Current Nodes
n_tf1=3641
n_tf2=4730
n_tf3=5819
n_tf4=6908
n_tf5=7997
n_tf6=9086
/com Modular Coil Current Nodes
n_m11=13599
n_m12=13609
n_m13=13619
n_m14=15619
n_m21=21639
n_m22=21649
n_m23=21659
n_m24=23659
n_m31=29679
n_m32=29689
n_m33=29699
```

```
n_m34=31699
/com Make nodal components out of Modular Coil Current Nodes
*do,jj,1,n_mcoils
nsel,none
*do,j,1,4
nsel,a,node,,n_m%jj%%j%
*enddo
cm,n_m%jj%,node
*enddo

/com
/com Define Coil Currents
/com
*dim,i_coils,array,6,12   ! Time,M1,M2,M3,PF1,PF2,PF3,PF4,PF5,PF6,TF,Plasma
/com Proposed Worst 6 Cases (from Wayne's TDS_XL_C08R00_c3.pdf)
i_coils(1,1)=1,2,3,4,5,6
i_coils(1,2)=0,38141,40908,34200,34200,695e3/20
i_coils(1,3)=0,35504,41561,32057,32057,706e3/20
i_coils(1,4)=0,35453,40598,32184,32184,621e3/18
i_coils(1,5)=0,-25123,-15274,11354,21858,0
i_coils(1,6)=0,-25123,-15274,11354,21858,0
i_coils(1,7)=0,-9698,-5857,-11802,-5975,160e3/72
i_coils(1,8)=0,-7752,-9362,-13936,-9441,-192e3/80
i_coils(1,9)=0,8284,1080,4563,4634,24.2e3/24
i_coils(1,10)=0,-8997,-24,5068,5705,10.7e3/14
i_coils(1,11)=16200,-3548,-1301,2191,2191,194.4e3/12
!i_coils(1,11)=0,-3548,-1301,2191,2191,194.4e3/12
i_coils(1,12)=0,0,0,-320775,-320775,0

/com
/com Modular Coils (two WP per coil, so two F commands per coil)
/com PDR analysis has 2x as much current as it should have
/com
tpmc1=10*1      ! turns per MC1 winding pack, 4 in-hand
tpmc2=10*1      ! turns per MC2 winding pack, 4 in-hand
tpmc3=9*1       ! turns per MC3 winding pack, 4 in-hand

*do,j,1,n_mcoils
cmsel,s,n_m%j%
f,all,amps,(tpmc%j%)*i_coils(time_pt,j+1)
*enddo

/com PF Coils
tppf1=72       ! turns per PF1 winding
tppf2=72       ! turns per PF2 winding
tppf3=72       ! turns per PF3 winding
tppf4=80       ! turns per PF4 winding
tppf5=24       ! turns per PF5 winding
tppf6=14       ! turns per PF6 winding
*do,j,1,n_pfcoils
nsel,s,node,,n_pf%j%u
nsel,a,node,,n_pf%j%l
f,all,amps,(tppf%j%)*i_coils(time_pt,j+4)
*enddo

/com TF Coils
tptf=12
*do,j,1,n_tfcoils
nsel,s,node,,n_tf%j%
f,all,amps,tptf*i_coils(time_pt,11)
*enddo

/com Plasma Current in Type,2 Real,1
r,1,2,i_coils(time_pt,12),0.03,0.05


/com
/com Make a ring of elements on the Plasma Axis
/com
*if,time_pt,eq,1,then
csys
```

```
wpcsys
vsel,none
esel,none
nsel,none
cylind,r0-0.005,r0+0.005,-0.005,0.005,-60,60
vatt,7
esize,0.1
type,3
mp,murx,7,1
vmesh,all
d,all,mag
d,all,temp
d,all,volt
*endif

/com
/com Change Width of TF WP
/com
h_wr=0.731       ! height of wedged region
w_wp=3.47*k      ! width of narrow WP (set to zero to maintain reference WP width)
w_wp=3.73*k      ! width of narrow WP w/0.250" GW
w_wp=3.98*k      ! width of narrow WP w/0.375" GW
th_wr=40         ! angular extent of wedged region
*if,w_wp,gt,0,then
esel,s,mat,,2
nsle
csys,1
nsel,r,loc,x,,r0
nsel,r,loc,z,-h_wr-t,h_wr+t
cm,n_wr,node
shpp,off
*do,j,1,6
local,100+j,1,,,,-50+(j-1)*20
esel,s,mat,,2
nsle
nsel,r,loc,y,-12,12
esln,,1
nsle
cmsel,u,n_wr
nsel,u,loc,y,-.1,.1
cm,nj_fix,node
/com -Y Nodes
local,110+j,,,,,-50+(j-1)*20
nsel,r,loc,y,-12,t
nmodif,all,,-w_wp/2
/com Upper Arc of -Y Nodes
csys,10+8*(j)
nsel,r,loc,y,180-th_wr,180
csys,1
nmodif,all,,-60+(j-1)*20
/com Lower Arc of -Y Nodes
cmsel,s,nj_fix
csys,110+j
nsel,r,loc,y,-12,t
csys,11+8*(j)
nsel,r,loc,y,-180,th_wr-180
csys,1
nmodif,all,,-60+(j-1)*20
/com +Y Nodes
cmsel,s,nj_fix
csys,110+j
nsel,r,loc,y,-t,12
nmodif,all,,+w_wp/2
/com Upper Arc of +Y Nodes
csys,10+8*(j)
nsel,r,loc,y,180-th_wr,180
csys,1
nmodif,all,,-60+(j)*20
/com Lower Arc of +Y Nodes
cmsel,s,nj_fix
csys,110+j
```

```
nsel,r,loc,y,-t,12
csys,11+8*(j)
nsel,r,loc,y,-180,th_wr-180
csys,1
nmodif,all,,-60+(j)*20
*enddo
esel,s,mat,,2
nsle
csys,1
nsel,r,loc,x,,r0
numm,node
*endif



itf=i_coils(time_pt,11)
tme=i_coils(time_pt,1)
/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%
allsel
save
fini

/com
/com Pull in Structural Solid
/com
vsel,none
asel,none
lsel,none
ksel,none
/AUX15
IOPTN,IGES,NODEFEAT
IOPTN,MERGE,YES
IOPTN,SOLID,YES
IOPTN,SMALL,YES
IOPTN,GTOLER, DEFA
IGESIN,'inboard_beams_steps_inter_1coiltop_LM10_asm','igs',' '
fini

/prep7
/com Make structural element
*get,etmx,etyp,,num,max
et,etmx+1,92
et,etmx+2,200,5
/com Mesh Structure
cel=2*k
esize,cel
/com Mesh Inner Volume Faces
csys,1
asel,s,loc,x,,r0
asel,r,loc,z,.5,12
allsel,below,area
*get,thmn_inface,kp,,mnloc,y
*get,thmx_inface,kp,,mxloc,y
asel,r,loc,y,thmn_inface-0.1,thmn_inface+0.1
cm,a1_inface,area
asll,,1
asel,r,loc,y,thmx_inface-0.1,thmx_inface+0.1
cm,a2_inface,area
cmsel,s,a1_inface
type,etmx+2
amesh,all
asll,,1
mshcopy,area,a1_inface,a2_inface,1,,(thmx_inface-thmn_inface),,t
/com Mesh Outer Volume Faces
csys,1
asel,s,loc,x,r0,12
asel,r,loc,z,.5,12
allsel,below,area
*get,thmn_outface,kp,,mnloc,y
*get,thmx_outface,kp,,mxloc,y
asel,r,loc,y,thmn_outface-0.1,thmn_outface+0.1
```

```
cm,a1_outface,area
asll,,1
asel,r,loc,y,thmx_outface-0.1,thmx_outface+0.1
cm,a2_outface,area
cmsel,s,a1_outface
type,etmx+2
amesh,all
asll,,1
mshcopy,area,a1_outface,a2_outface,1,,(thmx_outface-thmn_outface),,t
/com Mesh all volumes
allsel
type,etmx+1
vmesh,all

/com Generate the other upper structural set
csys,1
vsel,u,mat,,7
VGEN,2,all,,,,60,,,0
/com Reflect to get the bottom structure
csys
vsymm,z,all

/com Merge Adjacent Nodes
/com Outboard Interface
esel,s,type,,etmx+1
nsle
cm,struc_node,node
nsel,none
cm,merge_node,node
*do,j,1,5
cmsel,s,struc_node
csys
wpcsys
wprot,(j-1)*20-40
cswpla,200
nsel,r,loc,y,-0.10*k,0.10*k
nsel,r,loc,x,r0,12
cmsel,a,merge_node
cm,merge_node,node
*enddo
cmsel,s,merge_node
numm,node,0.050*k*1.25
/com Inboard Interface
esel,s,type,,etmx+1
nsle
csys
nsel,r,loc,y,-0.10*k,0.10*k
nsel,r,loc,x,,r0
numm,node,0.050*k*1.25
nsle
numcmp,node

/com Write Structural Coupling for Cyclic Symmetry
allsel
*get,ncp,cp,,num,max
modmesh,detach
esel,s,type,,1
esel,a,type,,etmx+1
nsle
csys,1
nrotate,all
esel,s,type,,etmx+1
nsle
nsel,r,loc,y,-60.1,-59.9
cm,neg60,node
nsle
nsel,r,loc,y,59.9,60.1
cm,pos60,node
esel,s,mat,,1,2
nsle
nsel,r,loc,y,-60.005,-59.995
```

```
cmsel,a,neg60
cm,neg60,node
nsle
nsel,r,loc,y,59.995,60.005
cmsel,a,pos60
cm,pos60,node

/com The cpcyc commands take too long unless the wedge face nodes are the only ones selected
csys,1
esel,s,mat,,1,2
esel,r,type,,1,3,2
nsle
cmsel,r,pos60
cm,ncoils1,node
esel,s,mat,,1,2
esel,r,type,,1,3,2
nsle
cmsel,r,neg60
cmsel,a,ncoils1
nrotate,all
cpcyc,ux,0.1*k,1,,120,,1
cpcyc,uy,0.1*k,1,,120,,1
cpcyc,uz,0.1*k,1,,120,,1
esel,s,mat,,10
esel,r,type,,4
nsle
cmsel,r,pos60
cm,nstruc1,node
esel,s,mat,,10
esel,r,type,,4
nsle
cmsel,r,neg60
cmsel,a,nstruc1
nrotate,all
cpcyc,ux,0.1*k,1,,120,,1
cpcyc,uy,0.1*k,1,,120,,1
cpcyc,uz,0.1*k,1,,120,,1

/com
/com Couple UX/UY/UZ at TF outboard equator
/com
esel,s,mat,,2
nsle
csys,1
nsel,r,loc,z
nsel,r,loc,x,r0,12
cpint,ux
cpint,uy
cpint,uz

/com
/com Constrain sides of WP to TF Case Side-Wall Shims
/com
dz_tfcoil=0.135
*if,w_wp,gt,0,then
dz_tfcoil=w_wp
*endif
!dz_tfcase=0.153     ! not used
*do,j,1,6
csys,1
esel,s,mat,,2,11,9
nsle
nsel,r,loc,y,-60+(j-1)*20,-60+(j-0)*20
nsel,u,loc,z,-0.75,0.75
esln,,1
csys,13+8*(j-1)
nsel,u,loc,z,-dz_tfcoil/3,dz_tfcoil/3
csys,10+8*j
nsel,u,loc,y,180-th_wr,180
csys,11+8*j
nsel,u,loc,y,-180,th_wr-180
```

```
csys,13+8*(j-1)
nrotate,all
nsle
cm,n_temp,node
cm,e_temp,elem
esel,r,mat,,11
cm,e_keep,elem
esln,,1
esel,r,mat,,2
nsle
nsel,u,loc,z,-dz_tfcoil/4,dz_tfcoil/4
csys,10+8*j
nsel,u,loc,y,180-th_wr,180
csys,11+8*j
nsel,u,loc,y,-180,th_wr-180

cm,n_keep,node
cmsel,s,n_keep
cmsel,s,e_keep
*if,w_wp,eq,0,then
ceint,,uz
*else
ceint,2.0,uz      ! need a big tolerance for when w_wp < TF Shim opening
*endif
*enddo


/com
/com Constrain midplane of WP (only good for TF On Load Case)
/com
*if,time_pt,eq,1,then
*do,j,1,6
csys,1
esel,s,mat,,2
nsle
csys,13+8*(j-1)
nsel,r,loc,z,-dz_tfcoil/3,dz_tfcoil/3
!nrotate,all
!d,all,uz
*enddo
*endif

/com
/com Couple the Ring Coils to their supports
/com
esel,s,mat,,1
esel,r,type,,1,3,2
nsle
csys,1
nrotate,all
/com CS Radii
esel,s,mat,,9
nsle
csys,1
*get,ro_cs,node,,mxloc,x
*get,ri_cs,node,,mnloc,x

/com PF4
/com
zsup_pf4=1.4478
esel,s,mat,,1
esel,r,type,,1
nsle
csys,1
nsel,r,loc,x,ro_cs+0.001,r0            ! PF4 must be inboard of R0
esln,,1
cm,pf4_elem,elem            ! all PF4 elements
nsel,r,loc,z,,12
esln,,1
cm,pf4u_elem,elem           ! PF4Upper elements
nrotate,all
```

```
*get,zbot_pf4,node,,mnloc,z
*get,ro_pf4,node,,mxloc,x
*get,ri_pf4,node,,mnloc,x
esel,invert
cmsel,r,pf4_elem
cm,pf4l_elem,elem           ! PF4Lower elements
nrotate,all
/com Set Up Constraints on PF4U
esel,s,mat,,10
nsle
nsel,r,loc,z,zsup_pf4-t,zsup_pf4+t
nsel,r,loc,x,ri_pf4-cel/4,ro_pf4+cel/4
nrotate,all
cmsel,s,pf4u_elem
ceint,,uz
ceint,,uy
/com Set Up Constraints on PF4L
esel,s,mat,,10
nsle
nsel,r,loc,z,-zsup_pf4-t,-zsup_pf4+t
nsel,r,loc,x,ri_pf4-cel/4,ro_pf4+cel/4
cmsel,s,pf4l_elem
ceint,,uz
ceint,,uy

/com
/com PF5
/com
esel,s,mat,,1
esel,r,type,,1
nsle
csys,1
nsel,u,loc,x,,r0
*get,ri_pf5,node,,mnloc,x
nsel,r,loc,x,ri_pf5-t,ri_pf5+t
nsel,r,loc,z,,12
*get,zbot_pf5,node,,mnloc,z
nsle
nsel,r,loc,x,ri_pf5-t,12

nsel,r,loc,z,zbot_pf5-t,,zbot_pf5+t
*get,ro_pf5,node,,mxloc,x
nsle
nsel,r,loc,x,ri_pf5-t,ro_pf5+t
esln,,1
nrotate,all
cm,pf5_elem,elem
nsel,r,loc,z,,12
esln,,1
cm,pf5u_elem,elem
esel,invert
cmsel,r,pf5_elem
cm,pf5l_elem,elem
/com Set Up Constraints on PF5U
esel,s,mat,,10
nsle
nsel,r,loc,z,zbot_pf5-t,zbot_pf5+t
nsel,r,loc,x,ri_pf5,ro_pf5
nrotate,all
cmsel,s,pf5u_elem
ceint,,uz
ceint,,uy
/com Set Up Constraints on PF5L
esel,s,mat,,10
nsle
nsel,r,loc,z,-zbot_pf5-t,-zbot_pf5+t
nsel,r,loc,x,ri_pf5,ro_pf5
nrotate,all
cmsel,s,pf5l_elem
ceint,,uz
ceint,,uy
```

```
/com
/com PF6
/com
esel,s,mat,,1
esel,r,type,,1
cmsel,u,pf4_elem
cmsel,u,pf5_elem
nsle
csys,1
nsel,u,loc,x,,ro_cs+0.001
esln,,1
cm,pf6_elem,elem
nsle
csys,1
nrotate,all
*get,ri_pf6,node,,mnloc,x
*get,ro_pf6,node,,mxloc,x
*get,zbot_pf6,node,,mxloc,z
nsel,r,loc,z,,12
esln,,1
cm,pf6u_elem,elem
esel,invert
cmsel,r,pf6_elem
cm,pf6l_elem,elem
/com Set Up Constraints on PF6U
esel,s,mat,,10
nsle
nsel,r,loc,z,zbot_pf6-t,zbot_pf6+t
nsel,r,loc,x,ri_pf6,ro_pf6
cmsel,s,pf6u_elem
ceint,,uz
ceint,,uy
/com Set Up Constraints on PF6L
esel,s,mat,,10
nsle
nsel,r,loc,z,-zbot_pf6-t,-zbot_pf6+t
nsel,r,loc,x,ri_pf6,ro_pf6
cmsel,s,pf6l_elem
ceint,,uz
ceint,,uy

/com
/com Make a component out of the CS (call it PF3 for logic reasons)
/com
esel,s,type,,1
nsle
csys,1
nsel,r,loc,x,ri_cs-0.001,ro_cs+0.001
esln,,1
cm,cs_elem,elem
cm,pf3_elem,elem
nsel,r,loc,z,,12
esln,,1
cm,pf3u_elem,elem
esel,invert
cmsel,r,pf3_elem
cm,pf3l_elem,elem


/com
/com Restrain The TF Coil in the Global Z
/com
*do,j,1,6
esel,s,mat,,2
nsle
csys,13+8*(j-1)
nsel,r,loc,z
csys,1
*get,zmn2%j%,node,,mnloc,z
nsel,r,loc,z,zmn2%j%-t,zmn2%j%+t
```

```
*get,thn2%j%,node,,mnloc,y
*get,r2%j%,node,,mnloc,x
nsel,r,node,,node(r2%j%,thn2%j%,zmn2%j%)
csys,1
nrotate,all
d,all,uz
*enddo

/com
/com Restrain the CS in Z and theta at the bottom
/com
cmsel,s,cs_elem
nsle
*get,zmn_cs,node,,mnloc,z
nsel,r,loc,z,zmn_cs
nsel,r,loc,y
d,all,uy,,,,,uz

/com
/com Restrain the TF Structure in Z & theta (Function of Temperature)
/com
/com At the inner shelf/flange/flats
/com
zin_struc=1.38938   ! Elevation of flat
r1in_struc=0.90     ! outer radius of flat
r2in_struc=1.05     ! inner radius of flat
thin_struc=60       ! angular frequency of supports
dthin_struc=26      ! angular extent of supports
nxdthin_struc=12    ! angular extent of nixed supports
nsel,none
cm,nin_struc,node
*do,j,1,2
esel,s,mat,,10
nsle
csys,1
nsel,r,loc,z,-zin_struc-t,zin_struc+t
nsel,r,loc,x,r1in_struc,r2in_struc
nsel,r,loc,y,(j-1)*thin_struc-30-dthin_struc/2,(j-1)*thin_struc-30+dthin_struc/2
nsel,u,loc,y,(j-1)*thin_struc-30-nxdthin_struc/2,(j-1)*thin_struc-30+nxdthin_struc/2
cmsel,a,nin_struc
cm,nin_struc,node
*enddo
cmsel,s,nin_struc
nrotate,all
d,all,uy!,,,,,uz
*get,nstrt,node,,num,min
*get,nstop,node,,num,max
*do,j,nstrt,nstop,1
*get,nselj,node,j,nsel
*if,nselj,ne,1,cycle
d,j,ux,nx(j)*alpha_struc*(tmp-t_ref)
d,j,uz,nz(j)*alpha_struc*(tmp-t_ref)
*enddo
/com
/com At the outer shelf/flange/flats
/com
zout_struc=1.246691  ! Elevation of flat
r1out_struc=1.98     ! outer radius of flat
r2out_struc=2.05     ! inner radius of flat
thout_struc=20       ! angular frequency of supports
dthout_struc=16      ! angular extent of supports
nxdthout_struc=7.5   ! angular extent of nixed supports

nsel,none
cm,nout_struc,node
*do,j,1,6
esel,s,mat,,10
nsle
csys,1
nsel,r,loc,z,-zout_struc-t,zout_struc+t
nsel,r,loc,x,r1out_struc,r2out_struc
```

```
nsel,r,loc,y,(j-1)*thout_struc-50-dthout_struc/2,(j-1)*thout_struc-50+dthout_struc/2
nsel,u,loc,y,(j-1)*thout_struc-50-nxdthout_struc/2,(j-1)*thout_struc-50+nxdthout_struc/2
cmsel,a,nout_struc
cm,nout_struc,node
*enddo
cmsel,s,nout_struc
nrotate,all
d,all,uy!,,,,,uz
*get,nstrt,node,,num,min
*get,nstop,node,,num,max
*do,j,nstrt,nstop,1
*get,nselj,node,j,nsel
*if,nselj,ne,1,cycle
d,j,ux,nx(j)*alpha_struc*(tmp-t_ref)
d,j,uz,nz(j)*alpha_struc*(tmp-t_ref)
*enddo

/com
/com Nix CEs
/com
!esel,all
!nsle
!cedele,all

/com Nix 200 elements
esel,s,type,,etmx+2
!edele,all

/com
/com Mag BCs
/com
esel,s,type,,1,3,2
nsle
d,all,mag

/com
/com Thermal BCs
/com
/com CS Temp
cmsel,s,cs_elem
nsle
d,all,temp,tmp
/com PF4 Temp
cmsel,s,pf4_elem
nsle
d,all,temp,tmp
/com PF5 Temp
cmsel,s,pf5_elem
nsle
d,all,temp,tmp
/com PF6 Temp
cmsel,s,pf6_elem
nsle
d,all,temp,tmp
/com TF Temp
esel,s,mat,,2
nsle
d,all,temp,tmp
/com Mod Coils
esel,s,mat,,3,5
nsle
d,all,temp,tmp

/com Oh, Just do it to all TEMP nodes
esel,s,type,,1,3,2
nsle
d,all,temp,tmp

/com Structure Temp
allsel
bfunif,temp,tmp
```

```
allsel
save
fini

:100
/solu
cnvtol,f,1e8,0.99             ! do not let the program try to converge on F
/com Solve Conduction Problem
allsel
solve

/com Solve for the fields on these nodes...
esel,s,mat,,1,2     ! PFs and TFs
esel,a,mat,,3,5    ! M1/2/3
esel,a,mat,,7       ! R0
esel,r,type,,1,3,2
nsle
/com ...from these elements
esel,s,mat,,1,5
esel,r,type,,1,3
esel,a,type,,2
biot,new

/com
/com Solve for the Stresses
/com
/com Apply a pressure to the back leg to simulate the radial load from the 2x4000 lb preload
/com
z_preload=52*k           ! max vertical location of applied preload
dth_preload=20           ! local angular extent of preload
r_ulback=1.322           ! Back-Side radius of upper and lower arcs
r_midback=2.81           ! Equatorial plane radius of back side

*do,j,1,6
nsel,none
cm,npre%j%,node
esel,s,mat,,2
nsle
csys,1
nsel,r,loc,x,r0,12
nsel,r,loc,y,-60+(j-1)*20,-60+(j-0)*20
nsel,r,loc,z,-z_preload,z_preload
cm,temp%j%,node
/com Upper Arc
csys,13+8*(j-1)
nsel,r,loc,x,r_ulback-0.01,r_ulback+0.01
nsel,r,loc,y,,180
*get,thmxpre,node,,mxloc,y
nsel,r,loc,y,thmxpre-dth_preload,180
cmsel,a,npre%j%
cm,npre%j%,node
/com Lower Arc
cmsel,s,temp%j%
csys,14+8*(j-1)
nsel,r,loc,x,r_ulback-0.01,r_ulback+0.01
nsel,r,loc,y,-180,0
*get,thmnpre,node,,mnloc,y
nsel,r,loc,y,-180,thmnpre+dth_preload
cmsel,a,npre%j%
cm,npre%j%,node
/com Middle (do not apply any load here)
cmsel,s,temp%j%
csys,1
nsel,r,loc,x,r_midback-0.01,r_midback+0.01
!cmsel,a,npre%j%
!cm,npre%j%,node
/com Apply Nodal Forces to these nodes (will not work because of rotated WP nodes for CEs)
!cmsel,s,npre%j%
!*get,n_pre,node,,count
!*afun,deg
```

```
!fcum,add
!f,all,fx,-(2*f_preload/n_pre)*cos(-50+(j-1)*20)
!f,all,fy,-(2*f_preload/n_pre)*sin(-50+(j-1)*20)
/com Apply a surface Pressure to these nodes
cmsel,s,npre%j%
esln
sf,all,pres,31.5*f_preload     ! scale factor gets correct radial load from applied pressure
*enddo


allsel
solve
fini
:1000
/post1
/dev,font,1,Arial,700,0,-19,0,0,,,
/view,1,1,1,1
/vup,1,z
/auto
!set,last
/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%
esel,s,mat,,2
nsle
plns,volt
plns,b,sum
plns,s,eqv
plns,u,z
plns,u,sum
esel,s,mat,,1,5
esel,r,type,,1,3,2
nsle
plns,volt
esel,s,type,,1,4,3
nsle
plns,u,sum
esel,s,mat,,10,11
nsle
plns,s,int
esel,s,mat,,2
nsle
csys,1
nsel,r,loc,x,,r0
nsel,r,loc,z,-.25,.25
esln,,1
/view
/type,1,0
/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%, TF Inboard Legs (+/-0.25 m) F=JxBdV
PLVECT,FMAG, , , ,VECT,ELEM,ON,0
/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%, TF Inboard Legs (+/-0.25 m) von Mises
/view,1,1,,1
/type
plns,s,eqv

/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%, CS von Mises
cmsel,s,cs_elem
nsle
rsys,1
plns,s,eqv
plns,s,z
plns,s,y
rsys

/com Sum forces on Ring coils
esel,s,mat,,10
nsle
csys,1
*get,thmn,node,,mnloc,y
*get,thmx,node,,mxloc,y
zsym=nint(360/(thmx-thmn))
*do,j,3,6
rsys,1
```

```
cmsel,s,pf%j%_elem
etab,fz,f,z
cmsel,s,pf%j%u_elem
ssum
*get,fz%j%u,ssum,,item,fz
fu=nint(zsym*fz%j%u/1000)
cmsel,s,pf%j%l_elem
ssum
*get,fz%j%l,ssum,,item,fz
fl=nint(zsym*fz%j%l/1000)

cmsel,s,pf%j%_elem
esel,u,mat,,9
nsle
/type,1,0
/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%, Fz(PF%j%U/L)=%fu%/%fl% kN
PLVECT,FMAG, , , ,VECT,ELEM,ON,0
/type
/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%, PF%j% von Mises, Fz(U/L)=%fu%/%fl%kN
plns,s,eqv
*enddo

/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%, Force Density (N/mm**3)
/com Force Density
/triad,off
esel,s,type,,1,3,2
esel,u,mat,,9
nsle
etab,fmag,fmag,sum
etab,volu,volu
SEXP,fd,FMAG,VOLU,1,-1,
sadd,mfd,fd,,1e-6
/view,1,-1,1,.25
esel,s,mat,,1
esel,r,type,,1,3,2
nsle
plet,mfd,avg
esel,s,mat,,2
nsle
/view,1,1,0.4,.25
plet,mfd,avg
esel,s,mat,,3,5
nsle
/view,1,1
plet,mfd,avg

/title,pdrcoils1%rn% I(TF)=%itf%A, LC=%time_pt%
esel,s,mat,,2
nsle
plns,s,int

allsel

:1001
/com List element centroids and their displacements
!/post1
allsel
ETABLE,ERASE
*get,ls,active,,set,lstp
/stitle,1,pdrcoils1%rn%, Load Step %ls%
esel,s,mat,,2
nsle
*get,e_strt,elem,,num,min
*get,e_stop,elem,,num,max
*get,e_num,elem,,count
*dim,e_cent,,e_num,7
csys
rsys
etab,u_x,u,x
etab,u_y,u,y
etab,u_z,u,z
```

```
count=0
*do,j,e_strt,e_stop
*if,esel(j),ne,1,cycle
count=count+1
e_cent(count,1)=count
*get,ejx,elem,j,cent,x
*get,ejy,elem,j,cent,y
*get,ejz,elem,j,cent,z
*get,uxj,elem,j,etab,u_x
*get,uyj,elem,j,etab,u_y
*get,uzj,elem,j,etab,u_z
e_cent(count,2)=ejx
e_cent(count,3)=ejy
e_cent(count,4)=ejz
e_cent(count,5)=uxj
e_cent(count,6)=uyj
e_cent(count,7)=uzj
*enddo
tp=time_pt
/output,tflocdisp%rn%LC%tp%LS%ls%,lis
*vwrite,
('  Element #        X            Y            Z           UX           UY           UZ')
*vwrite,e_cent(1,1),e_cent(1,2),e_cent(1,3),e_cent(1,4),e_cent(1,5),e_cent(1,6),e_cent(1,7)
(1p7e13.5)
/output
/eof
/type
fini
/exit,all
/eof
```